> restart;
> with(LinearAlgebra):
> Digits := 18:
y'' = p(x)*y'+q(x)*y+r(x) on [a,b] with y(a)=alpha and y(b)=beta

An example with g a solution

> a := 1: alpha := evalf(1/2):

b := 2: beta := evalf(ln(2)):

g := x -> 4/x-2/x^2+ln(x)-3/2:  
> p:= x -> -4/x:  
> q := x -> -2/x^2:  
> r := x -> 2*ln(x)/x^2:  
> expand(-diff(g(x),x$2)+p(x)*diff(g(x),x)+q(x)*g(x)+r(x)); 

> FDiffSolve := proc(p,q,r,a,b,alpha,beta,N)

   # "solve y''=p(x)*y'+q(x)*y+r(x) on [a,b]";

   # with y(a)=alpha, y(b)=beta, h= (b-a)/(N+1)";

local A:= Matrix(N):

local xx:= Vector(N): 

local w := Vector(N): 

local bb := Vector(N):

local h:

local j:

h:= evalf((b-a)/(N+1)):

for j from 1 to N do 

xx[j] := a+ j*h: 

A[j,j] := 2+h^2*q(xx[j]): 

od:

for j from 1 to N-1 do

A[j,j+1] := -1 + p(xx[j])*h/2;

od:

for j from 2 to N do

A[j,j-1] := -1- p(xx[j])*h/2;

od:

bb[1] := -h^2*r(xx[1])+(1+h/2*p(xx[1]))*alpha:

bb[N] := -h^2*r(xx[N])+(1-h/2*p(xx[N]))*beta:

for j from 2 to N-1 do

bb[j] := -h^2*r(xx[j]);

od:

LinearSolve(A,bb):

end proc;

> N:= 10:

for k from 0 to 8 do

NN:=(N+1)*2^k-1:

w:= Vector(NN):

ww:=Vector(NN):

xx:= Vector(NN):

w:= FDiffSolve(p,q,r,a,b,alpha,beta,NN):

h:= evalf((b-a)/(NN+1)):

for j from 1 to NN do

xx[j]:= a+h*j:

ww[j] := w[j]-g(xx[j]):

od:

print(NN,Norm(ww));

od: 

Note that the accuracy going down by about 1/4

> 

Let's try a simple Richardson extrapolation.

> N:= 10:

K:= 5:

for k from 0 to K+1 do

NN:=(N+1)*2^k-1:

w||k:= Vector(NN):

ww||k:=Vector(NN):

xx||k:= Vector(NN):

w||k:= FDiffSolve(p,q,r,a,b,alpha,beta,NN):

h:= evalf((b-a)/(NN+1)):

for j from 1 to NN do

xx||k[j]:= a+h*j:

ww||k[j] := w||k[j]-g(xx||k[j]):

od:

od: 
> 

> for k from 0 to K do

NN:=(N+1)*2^k-1:

Richardson||(k+1) := Vector(NN):

Z||k:=Vector(NN):

for j from 1 to NN do

  Richardson||(k+1)[j] := (-w||k[j]+4*w||(k+1)[2*j])/3:

Z||k[j] := Richardson||(k+1)[j]-g(xx||k[j]):

od:

od:

> for k from 0 to K-1 do

print(NN,Norm(ww||k),Norm(ww||k)/Norm(ww||(k+1))):

print(Norm(Z||k),Norm(Z||k)/Norm(Z||(k+1)));

od:

